
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.2.3.14, which is part of the LTE test suite.
The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_7_2_3_14
Test Group:
MAC
ATS Version:
iwd-EUTRA-TVB2009-03_D09wk51
System Simulator used:
Anite CT
UE used:
LG Electronics LEO3 UE.

Verification Status:
PASS
4 Corrections required for test case 7.2.3.14
4.1 Introduction

This section describes the changes required to make test case 7.2.3.14 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D09wk51 release.

4.2 Change 1
	Testcase name
	f_TC_7_2_3_14_EUTRA

	Reason for change
	After step 9A, data corresponding to AMD PDUs with sequence numbers 5, 4, and 3 are transmitted in downlink. These data are expected to be looped back in a single AMD PDU in step13. Instead of retrieving current timing information and sending AMD PDU with sequence number 3 after a delay of 100ms, AMD PDUs with sequence numbers 5, 4, 3 should be sent after retrieving current timing and after waiting for 100ms.
If not done this way, the moment AMD PDU with sequence number 5 is received at UE, t-Reordering starts with default value of 80ms. By the time PDU with sequence number 4 and followed by PDU with sequence number 3 (after a delay of 100ms) are received, t-Reordering expires and a status report is triggered. Since uplink grant configuration is to be done later at step 12A with a delay of 60ms, as a result UE will start PRACH process. 

	Summary of change
	Retrieve current timing information after step 9A instead of step 12. AMD PDUs with sequence numbers 5, 4 and 3 are sent on subsequent subframes. 

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	  var integer i;

  var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var RLC_Status_Padding_Type     v_Padding_1bit := '0'B;

  var template (value) RLC_Status_NACK_List_Type   v_NACK_List;

  timer t_WatchDog := 5.0;

  var SubFrameTiming_Type v_Timing;

…

…

…

  // Step 9A: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 10: The SS transmits an AMD PDU to the UE.

  // This PDU contains the last part of SDU#6.

  // 70 octets

  p_RLC_Rec.AM_VTS := 5; // SN to be used for the next PDU

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 22, 70);

  f_TxAMD_OnePDU ( p_RLC_Rec,

                   cs_TimingInfo_Now,

                   tsc_P_NoPoll,

                   tsc_FI_EndOfSDU,

                   v_RLC_Data1 );

  // Step 11: The SS transmits an AMD PDU to the UE.

  // This PDU contains the 2nd part of SDU#5, and the 1st  part of SDU#6.

  // 2nd part of SDU#5 !!!

  // 90 respectively 20 + 2 octets

  p_RLC_Rec.AM_VTS := 4; // SN to be used for the next PDU

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 62, 90);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 22);

  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo_Now,

                     tsc_P_NoPoll,

                     tsc_FI_MiddleOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

  // Step 11A: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 12: The SS transmits an AMD PDU to the UE.

  // This PDU carries SDU#4 and the 1st part of SDU#5.

  // 80 + 2 respectively 60 + 2 octets

  //100 ms to wait for having an accurate reference to start

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 10) mod 1024;
  p_RLC_Rec.AM_VTS := 3; // SN to be used for the next PDU

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 82);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 62);

  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});


After: 

	  var integer i := 0, j := 0;
  var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var RLC_Status_Padding_Type     v_Padding_1bit := '0'B;

  var template (value) RLC_Status_NACK_List_Type   v_NACK_List;

  timer t_WatchDog := 5.0;

  var SubFrameTiming_Type v_Timing;

  var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode ;

  var integer SubframeOffset := 0;
  // determine RAT to make the timing correct later on

  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);
…

…

…

  // Step 9A: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 10) mod 1024;
  // Step 10: The SS transmits an AMD PDU to the UE.

  // This PDU contains the last part of SDU#6.

  // 70 octets

  p_RLC_Rec.AM_VTS := 5; // SN to be used for the next PDU

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 22, 70);
  i := 0; j := 0;

  if (v_EUTRA_FDD_TDD_Mode == TDD)

    {

      j := ((i+1) / 2);

      SubframeOffset := i + j * 3;

    }

    else // FDD

    {

      SubframeOffset := i;      

    }
  f_TxAMD_OnePDU ( p_RLC_Rec,

                   cs_TimingInfo(v_Timing.SFN.Number, SubframeOffset),
                   tsc_P_NoPoll,

                   tsc_FI_EndOfSDU,

                   v_RLC_Data1 );

  // Step 11: The SS transmits an AMD PDU to the UE.

  // This PDU contains the 2nd part of SDU#5, and the 1st  part of SDU#6.

  // 2nd part of SDU#5 !!!

  // 90 respectively 20 + 2 octets

  p_RLC_Rec.AM_VTS := 4; // SN to be used for the next PDU

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 62, 90);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 22);
  i := i + 1;

  if (v_EUTRA_FDD_TDD_Mode == TDD)

    {

      j := ((i+1) / 2);

      SubframeOffset := i + j * 3;

    }

    else // FDD

    {

      SubframeOffset := i;      

    }
  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo(v_Timing.SFN.Number, SubframeOffset),
                     tsc_P_NoPoll,

                     tsc_FI_MiddleOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

  // Step 11A: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 12: The SS transmits an AMD PDU to the UE.

  // This PDU carries SDU#4 and the 1st part of SDU#5.

  // 80 + 2 respectively 60 + 2 octets

  //100 ms to wait for having an accurate reference to start

  p_RLC_Rec.AM_VTS := 3; // SN to be used for the next PDU

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 82);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 62);
  i := i + 1;

  if (v_EUTRA_FDD_TDD_Mode == TDD)

    {

      j := ((i+1) / 2);

      SubframeOffset := i + j * 3;

    }

    else // FDD

    {

      SubframeOffset := i;      

    }
  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo(v_Timing.SFN.Number, SubframeOffset),

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});


4.3 Change 2
	Testcase name
	f_TC_7_2_3_14_EUTRA

	Reason for change
	In step 18 and step 19, data corresponding to AMD PDUs with sequence numbers 7 and 6 are transmitted in downlink. These data are expected to be looped back in a single AMD PDU in step 20. Instead of retrieving current timing information and sending AMD PDU with sequence number 6 after a delay of 100ms, AMD PDUs with sequence numbers 7 and 6 should be sent after retrieving current timing and after waiting for 100ms.

The reason is similar to as described in change 2 above.

	Summary of change
	Retrieve current timing information after step 18 instead of step 19. AMD PDUs with sequence numbers 7 and 6 are then sent on subsequent sub frames.

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	// Step 18: The SS transmits an AMD PDU to the UE.

  // This PDU contains SDU#8, and the 1st  part of SDU#9.

  // 120 + 2 respectively 30 + 2 octets

  p_RLC_Rec.AM_VTS := 7; // SN to be used for the next PDU

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 122);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 32);

  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo_Now,

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

  // Step 18A: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 19: The SS transmits an AMD PDU to the UE.

  // This PDU carries SDU#7.

  // 200 + 2 octets

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 10) mod 1024;
  p_RLC_Rec.AM_VTS := 6; // SN to be used for the next PDU

  f_TxAMD_SDU ( p_RLC_Rec,

                cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                tsc_P_NoPoll,

                tsc_RLC_SDU7 );

  // Step 19A: The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback SDU#7, SDU#8 and SDU#9.

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number ));




After: 

	// Step 18: The SS transmits an AMD PDU to the UE.

  // This PDU contains SDU#8, and the 1st  part of SDU#9.

  // 120 + 2 respectively 30 + 2 octets

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 10) mod 1024;
  p_RLC_Rec.AM_VTS := 7; // SN to be used for the next PDU

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 122);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 32);

  i := 0; j := 0;

  if (v_EUTRA_FDD_TDD_Mode == TDD)

    {

      j := ((i+1) / 2);

      SubframeOffset := i + j * 3;

    }

    else // FDD

    {

      SubframeOffset := i;      

    }
  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo(v_Timing.SFN.Number, SubframeOffset),
                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

  // Step 18A: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 19: The SS transmits an AMD PDU to the UE.

  // This PDU carries SDU#7.

  // 200 + 2 octets

  p_RLC_Rec.AM_VTS := 6; // SN to be used for the next PDU
  i := i + 1;

  if (v_EUTRA_FDD_TDD_Mode == TDD)

    {

      j := ((i+1) / 2);

      SubframeOffset := i + j * 3;

    }

    else // FDD

    {

      SubframeOffset := i;      

    }
  f_TxAMD_SDU ( p_RLC_Rec,

                cs_TimingInfo(v_Timing.SFN.Number, SubframeOffset),

                tsc_P_NoPoll,

                tsc_RLC_SDU7 );

  // Step 19A: The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback SDU#7, SDU#8 and SDU#9.

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number ));


5 Execution Log Files

5.1 LG Electronics LEO3 UE

The LG electronics LEO3 UE passed this test case on the Anite LTE solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file tc_7_2_3_14_LG-log.html:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

	[1]
	R5s100003: This archive comprises text format execution log file and the TTCN file.

