
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.2.3.15, which is part of the LTE test suite. 

The test case can be demonstrated to run with more than one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_7_2_3_15
Test Group:
RLC
ATS Version:
iwd-EUTRA-TVB2009-03_D10wk07
System Simulator used:
Anite Conformance Toolset
UE used:
LG Electronics LEO3 and Samsung Tanum.

Verification Status:
PASS
4 Corrections required for test case 7.2.3.15
4.1 Introduction

This section describes the changes required to make test case 7.2.3.15 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D10wk07 release.

4.2 Change 1

	Testcase name
	f_TC_7_2_3_15_EUTRA

	Reason for change
	In test body the default UL Grant is being disabled so after the test body again configuring the default UL Grant.

	Summary of change
	Configure the default UL Grant.

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	function f_TC_7_2_3_15_EUTRA()

runs on EUTRA_PTC {

  var RLC_SS_State v_RLC_Rec;

  //EUTRA initialisation

  f_EUTRA_Init(c1);

  // RLC global initialisation

  f_InitRLC_Record(v_RLC_Rec, AM_Mode);

  // RLC test case specific initialisation

  v_RLC_Rec.UL_PDCP_SDU_size := 98;

  // Cell configuration, with one AM DRB at the SS side

  f_EUTRA_CellConfig_Def (eutra_Cell1);

  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  //entering UE test state 3, default values with BSR and PHR disabled

  f_EUTRA_RbEst_Def_DisableBSR_PHR ( eutra_Cell1 );

  // reconfigure DRB locally to set the required test mode AM mapped to TM

  f_EUTRA_SS_DRB1_ConfigAM_TM_Mode ( eutra_Cell1);

  //closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA ( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_1DRB (v_RLC_Rec.UL_PDCP_SDU_size * 8, tsc_DRB1) );

  f_EUTRA_TestBody_Set(true);

  fl_EUTRA_7_2_3_15_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  // leaving UE test state 4

  f_OpenUE_TestLoopMode_Deactivate_TestMode ( eutra_Cell1 );

  //postamble

  f_EUTRA_Postamble (eutra_Cell1, RRC_CONNECTED, ATTACHED );

} // f_TC_7_2_3_15_EUTRA


After:

	function f_TC_7_2_3_15_EUTRA()

runs on EUTRA_PTC {

  var RLC_SS_State v_RLC_Rec;

  //EUTRA initialisation

  f_EUTRA_Init(c1);

  // RLC global initialisation

  f_InitRLC_Record(v_RLC_Rec, AM_Mode);

  // RLC test case specific initialisation

  v_RLC_Rec.UL_PDCP_SDU_size := 98;

  // Cell configuration, with one AM DRB at the SS side

  f_EUTRA_CellConfig_Def (eutra_Cell1);

  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  //entering UE test state 3, default values with BSR and PHR disabled

  f_EUTRA_RbEst_Def_DisableBSR_PHR ( eutra_Cell1 );

  // reconfigure DRB locally to set the required test mode AM mapped to TM

  f_EUTRA_SS_DRB1_ConfigAM_TM_Mode ( eutra_Cell1);

  //closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA ( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_1DRB (v_RLC_Rec.UL_PDCP_SDU_size * 8, tsc_DRB1) );

  f_EUTRA_TestBody_Set(true);

  fl_EUTRA_7_2_3_15_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1, cs_TimingInfo_Now ); 

  // leaving UE test state 4

  f_OpenUE_TestLoopMode_Deactivate_TestMode ( eutra_Cell1 );

  //postamble

  f_EUTRA_Postamble (eutra_Cell1, RRC_CONNECTED, ATTACHED );

} // f_TC_7_2_3_15_EUTRA


4.3 Change 2
	Testcase name
	fl_EUTRA_7_2_3_15_TestBody

	Reason for change
	1) TDD mode handling at test step 1, 2, 3, 4, 5, 6, 7, 9 and 11 is missing.
2) UE need to loop back multiple PDUs at step 11d, 11e, 12, 13, 25, 26, 27 and 28. So it required that cyclic grants used that allow UE to transmit 1 RLC SDU per RLC PDU.
3) At test step 1, 2, 3, 4, 5, 6, 7, 9 and 11 sending PDUs at specific frame and sub-frame to ensure that UE should have data to transmit at the time of receiving grant.
4) At test step 18 UE is sending the NACK. To ensure the expected NACK SS need to transmit all required PDUs before expecting any PDU.
5) UE will transmit all the PDUs which are completely received before step 12.
6) Same status PDU sends two time at step 13a (ttcn) ,14 and 28a (ttcn) ,29.

7) At step 17, SS need to wait for T-reordering expiry before starting UL Grant.
8) As step 18 and 22 periodic cyclic grant can not be used as it not clear when UE will be ready for transmission of data in UL.

9) After step 24, SS should ensure that before sending the grant UE should have data to transmit.

	Summary of change
	1) Calculate and send PDU at the next DL sub-frame basis of sub-frame offset.
2) Stop the UL Grant and start grant when required (submitting prose CR for this).
3) Sending PDUs at specific frame and sub-frame and sending grant after 60 ms to ensure UE has data to transmit (Prose CR R5-100626).
4) Remove the Step 8, 10, 15 and 16 (Prose CR R5-100626).

5) Expecting SDU#1, SDU#2, SDU#3, and SDU #4. Step 11a, 11b, 11c, 11d and 11e (R5-100626).
6) Remove step 13a (ttcn) and 28a (ttcn).

7) At step 11c, SS start UL Grant after 80ms.
8) Use the UL Grant type on SR reception.

9) Adding delay of 60ms then allocating the grant (Prose CR R5-100626).

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	//----------------------------------------------------------------------------

function fl_EUTRA_7_2_3_15_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC {

  var integer i;

  var RLC_DataField_Type v_RLC_Data1;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var RLC_Status_Padding_Type     v_Padding_1bit := '0'B;

  var RLC_Status_Padding_Type     v_Padding_7bits := '0000000'B;

  var template (value) RLC_Status_NACK_List_Type   v_NACK_List;

  // Data size to be extracted from PRBS: 98

  p_RLC_Rec.TxDataSize := 98;

  //PDCP SDUs of this size will be built, appended to a PDCP header,

  //and then used as RLC SDUs

  // generate 8 RLC SDUs = PDCP PDUs

  for(i := 0; i < 8; i:= i + 1) {

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod (tsc_Maximum_PDCP_SN_12 + 1);

  }

  // Step 1: The SS transmits one AMD PDU containing SDU#8 (100 bytes) in its data field to the UE.

  // SN=7 indicates the loss of 7 PDUs.

  p_RLC_Rec.AM_VTS := 7;

  f_TxAMD_SDU ( p_RLC_Rec, cs_TimingInfo_Now, tsc_P_NoPoll, tsc_RLC_SDU8 );

  // Step 2: The SS transmits one AMD PDU segment containing 50 bvtes of SDU#1 in its data field to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#1, which contained SDU#1 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,
                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 3: The SS transmits one AMD PDU segment containing 50 bytes of SDU#2 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#2, which contained SDU#2 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 1;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,
                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 4: The SS transmits one AMD PDU segment containing 50 bytes of SDU#3 in its data field to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#3, which contained SDU#3 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 2;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,
                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 5: The SS transmits one AMD PDU segment containing 50 bytes of SDU#4 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#4, which contained SDU#4 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,
                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 6: The SS transmits one AMD PDU segment containing 50 bytes of SDU#4 in its data field to the UE.

  // This AMD PDU segment carries part 1of AMD PDU#4, which contained SDU#4 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,
                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 7: The SS transmits one AMD PDU segment containing 50 bytes of SDU#1 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#1, which contained SDU#1 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,
                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 8: Check: Does the UE transmit an RLC SDU containing SDU#1 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

  // Step 8a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);
  // Step 9: The SS transmits one AMD PDU segment containing 50 bytes of SDU#2 to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#2, which contained SDU#2 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 1;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,
                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 10: Check: Does the UE transmit an RLC SDU containing SDU#2 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU2 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

  // Step 10a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);
  // Step 11: The SS transmits one AMD PDU segment containing 50 bytes of SDU#3 in its data field to the UE.

  // This AMD PDU segment carries part 2 of PDU#3, which contained SDU#3 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 2;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,
                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 12: Check: Does the UE transmit an RLC SDU  containing SDU#3 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU3 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

  // Step 13: Check: Does the UE transmit an RLC SDU  containing SDU#4 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU4 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 13");

  // Step 13a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec); 
  // Step 14: The SS transmits an RLC STATUS PDU to the UE.

  // This PDU acks PDUs up to those including SDU#4. ACK_SN=4.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 15: The SS transmits one AMD PDU segment containing 50 bytes of SDU#7 in its data field to the UE.

  // This AMD PDU segment carries part  1 of PDU #7, which contained SDU#7 (100 bytes) in its data field. SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 16: The SS transmits one AMDPDU segment containing 50 bytes of SDU#6 in its data field to the UE.

  // This AMD PDU segment carries segment 2 of AMD PDU#6, which contained SDU#6 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 5;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);
  // Step 17: Wait t_Reordering to run out at the UE side.

  f_Delay (p_RLC_Rec.t_Reordering);
  // Step 18: Check: Does the UE transmit a Status Report with NACK_SN=4, NACK_SN=5 with SOStart=0 and SOEnd=49,

  // and  NACK_SN=6 with SOStart=50 and SOEnd=special value, and ACK_SN=8?

  p_RLC_Rec.AM_VTA := 8;

  v_NACK_List := cr_NACK_List_3NACKs ( 4,

                                       tsc_E2_NoSO_Follows,

                                       omit,

                                       5,

                                       tsc_E2_SO_Follows,

                                       cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(49, tsc_AM_SO_Size)),

                                       6,

                                       tsc_E2_SO_Follows,

                                       cr_Status_SO(int2bit(50, tsc_AM_SO_Size), tsc_SO_Special) );

  f_RxSTATUS_PDU_NACK ( p_RLC_Rec, v_NACK_List, v_Padding_1bit);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18");

  // Step 19: The SS transmits one AMD PDU segment containing 50 bytes of SDU#7 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#7, which contained SDU#7 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 20: The SS transmits one AMD PDU segment containing 50 bytes of SDU#6 in its data field to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#6, which contained SDU#6 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 5;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 21: The SS transmits one RLC PDU of 50 data bytes to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#5, which contained SDU#5 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 4;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 22: Wait t_Reordering to run out at the UE side

  f_Delay (p_RLC_Rec.t_Reordering);

  // Step 23: Check: Does the UE transmit a Status Report with NACK_SN=4 with SOStart=50

  // and SOEnd=32767 (special SOEnd value), and ACK_SN=8?

  p_RLC_Rec.AM_VTA := 8;

  v_NACK_List := cr_NACK_List ( 4,

                                tsc_E2_SO_Follows,

                                cr_Status_SO(int2bit(50, tsc_AM_SO_Size), tsc_SO_Special));

  f_RxSTATUS_PDU_NACK ( p_RLC_Rec, v_NACK_List, v_Padding_7bits );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 23");

  // Step 24: The SS transmits one AMD PDU containing 50 bytes of SDU#5 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#5, which contained SDU#5 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 4;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50, v_RLC_Data1);

  // Step 25: Check: Does the UE transmit an RLC SDU  containing SDU#5 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU5 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 25");

  // Step 26: Check: Does the UE transmit an RLC SDU  containing SDU#6 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU6 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 26");

  // Step 27: Check: Does the UE transmit an RLC SDU  containing SDU#7 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU7 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 27");

  // Step 28: Check: Does the UE transmit an RLC SDU  containing SDU#8 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU8 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 28");

  // Step 28a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);
  // Step 29: The SS transmits an RLC STATUS PDU to the UE.

  // This PDU acks PDUs up to those including SDU#7. ACK_SN=8.

  p_RLC_Rec.AM_VTA := 8;

  f_TxSTATUS_PDU ( p_RLC_Rec );

} // fl_EUTRA_7_2_3_15_TestBody


After:

	function fl_EUTRA_7_2_3_15_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC {

  var integer i;

  var RLC_DataField_Type v_RLC_Data1;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var RLC_Status_Padding_Type     v_Padding_1bit := '0'B;

  var RLC_Status_Padding_Type     v_Padding_7bits := '0000000'B;

  var template (value) RLC_Status_NACK_List_Type   v_NACK_List;

  var SubFrameTiming_Type v_Timing;

  var integer v_SubframeOffset;

  var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode; // NOTE 1 
  // Data size to be extracted from PRBS: 98

  p_RLC_Rec.TxDataSize := 98;

  //PDCP SDUs of this size will be built, appended to a PDCP header,

  //and then used as RLC SDUs

  // determine RAT to make the timing correct later on

  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1); // NOTE 1
  // generate 8 RLC SDUs = PDCP PDUs

  for(i := 0; i < 8; i:= i + 1) {

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod (tsc_Maximum_PDCP_SN_12 + 1);

  }

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now ); // NOTE 2
  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1); // NOTE 1
  // Step 1: The SS transmits one AMD PDU containing SDU#8 (100 bytes) in its data field to the UE.

  // SN=7 indicates the loss of 7 PDUs.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0); // NOTE 1
  p_RLC_Rec.AM_VTS := 7;

  f_TxAMD_SDU ( p_RLC_Rec, cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset), tsc_P_NoPoll, tsc_RLC_SDU8 ); // NOTE 3
  // Step 2: The SS transmits one AMD PDU segment containing 50 bvtes of SDU#1 in its data field to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#1, which contained SDU#1 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1); // NOTE 1
  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset), // NOTE 3
                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 3: The SS transmits one AMD PDU segment containing 50 bytes of SDU#2 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#2, which contained SDU#2 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2); // NOTE 1  

  p_RLC_Rec.AM_VTS := 1;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset), // NOTE 3
                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 4: The SS transmits one AMD PDU segment containing 50 bytes of SDU#3 in its data field to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#3, which contained SDU#3 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 3); // NOTE 1  

  p_RLC_Rec.AM_VTS := 2;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset), // NOTE 3
                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 5: The SS transmits one AMD PDU segment containing 50 bytes of SDU#4 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#4, which contained SDU#4 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 4);  

  p_RLC_Rec.AM_VTS := 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 50, 50); // NOTE 1
  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset), // NOTE 3
                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 6: The SS transmits one AMD PDU segment containing 50 bytes of SDU#4 in its data field to the UE.

  // This AMD PDU segment carries part 1of AMD PDU#4, which contained SDU#4 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 5);  // NOTE 1
  p_RLC_Rec.AM_VTS := 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset), // NOTE 3
                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 7: The SS transmits one AMD PDU segment containing 50 bytes of SDU#1 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#1, which contained SDU#1 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 6); // NOTE 1
  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset), // NOTE 3
                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 8: VOID    // NOTE 4
  // Step 9: The SS transmits one AMD PDU segment containing 50 bytes of SDU#2 to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#2, which contained SDU#2 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 7); // NOTE 1 

  p_RLC_Rec.AM_VTS := 1;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset), // NOTE 3
                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 10: VOID     // NOTE 4
  // Step 11: The SS transmits one AMD PDU segment containing 50 bytes of SDU#3 in its data field to the UE.

  // This AMD PDU segment carries part 2 of PDU#3, which contained SDU#3 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 8); // NOTE 1  

  p_RLC_Rec.AM_VTS := 2;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset),  // NOTE 3
                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 11a: The SS transmits one AMD PDU segment containing 50 bytes of SDU#7 in its data field to the UE.

  // This AMD PDU segment carries part  1 of PDU #7, which contained SDU#7 (100 bytes) in its data field. SO=0 and LSF=0.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 9);  

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset),

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 11b: The SS transmits one AMDPDU segment containing 50 bytes of SDU#6 in its data field to the UE.

  // This AMD PDU segment carries segment 2 of AMD PDU#6, which contained SDU#6 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  v_SubframeOffset := fl_CalcSubframeOffset(v_EUTRA_FDD_TDD_Mode, 10);  

  p_RLC_Rec.AM_VTS := 5;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset),

                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  v_Timing.Subframe.Number := 0;

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

   // Step 11c: The SS transmits an uplink grant allowing the UE to transmit 1 RLC SDU.

  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1,

                                             cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), 

                                             15, 3, 1, 4);

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 20);

  f_EUTRA_StartULGrantTransmissionOnSR_Reception( eutra_Cell1, 

                                                  cs_TimingInfo(v_Timing.SFN.Number , v_Timing.Subframe.Number), 

                                                  cs_DciInfo_CcchDcchDtchUL_Explicit( 15, 3 ) );

  // Step 11d: Check: Does the UE transmit an RLC SDU containing SDU#1 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU1 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11d");

  // Step 11e: Check: Does the UE transmit an RLC SDU containing SDU#2 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU2 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11e");                      // NOTE 5
  // Step 12: Check: Does the UE transmit an RLC SDU  containing SDU#3 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU3 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

  // Step 13: Check: Does the UE transmit an RLC SDU  containing SDU#4 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU4 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 13");

  // Step 13a: Send STATUS PDU upon Pollbit.

//  f_TxSTATUS_PDU(p_RLC_Rec);       // NOTE 6
  // Step 14: The SS transmits an RLC STATUS PDU to the UE.

  // This PDU acks PDUs up to those including SDU#4. ACK_SN=4.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 15: VOID         // NOTE 4
  // Step 16: VOID         // NOTE 4
  // Step 17: Wait t_Reordering to run out at the UE side.

//  f_Delay (p_RLC_Rec.t_Reordering);       // NOTE 7
  // Step 18: Check: Does the UE transmit a Status Report with NACK_SN=4, NACK_SN=5 with SOStart=0 and SOEnd=49,

  // and  NACK_SN=6 with SOStart=50 and SOEnd=special value, and ACK_SN=8?

  p_RLC_Rec.AM_VTA := 8;

  v_NACK_List := cr_NACK_List_3NACKs ( 4,

                                       tsc_E2_NoSO_Follows,

                                       omit,

                                       5,

                                       tsc_E2_SO_Follows,

                                       cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(49, tsc_AM_SO_Size)),

                                       6,

                                       tsc_E2_SO_Follows,

                                       cr_Status_SO(int2bit(50, tsc_AM_SO_Size), tsc_SO_Special) );

  f_RxSTATUS_PDU_NACK ( p_RLC_Rec, v_NACK_List, v_Padding_1bit);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18");

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now ); // NOTE 2
  // Step 19: The SS transmits one AMD PDU segment containing 50 bytes of SDU#7 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#7, which contained SDU#7 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 20: The SS transmits one AMD PDU segment containing 50 bytes of SDU#6 in its data field to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#6, which contained SDU#6 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 5;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 21: The SS transmits one RLC PDU of 50 data bytes to the UE.

  // This AMD PDU segment carries part 1 of AMD PDU#5, which contained SDU#5 (100 bytes) in its data field.

  // SO=0 and LSF=0.

  p_RLC_Rec.AM_VTS := 4;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 22: Wait t_Reordering to run out at the UE side

  f_Delay (p_RLC_Rec.t_Reordering);

  f_EUTRA_StartULGrantTransmissionOnSR_Reception( eutra_Cell1, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Explicit( 15, 3 ) );   // NOTE 8
  // Step 23: Check: Does the UE transmit a Status Report with NACK_SN=4 with SOStart=50

  // and SOEnd=32767 (special SOEnd value), and ACK_SN=8?

  p_RLC_Rec.AM_VTA := 8;

  v_NACK_List := cr_NACK_List ( 4,

                                tsc_E2_SO_Follows,

                                cr_Status_SO(int2bit(50, tsc_AM_SO_Size), tsc_SO_Special));

  f_RxSTATUS_PDU_NACK ( p_RLC_Rec, v_NACK_List, v_Padding_7bits );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 23");

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now ); // NOTE 2
  // Step 24: The SS transmits one AMD PDU containing 50 bytes of SDU#5 in its data field to the UE.

  // This AMD PDU segment carries part 2 of AMD PDU#5, which contained SDU#5 (100 bytes) in its data field.

  // SO=50 and LSF=1.

  p_RLC_Rec.AM_VTS := 4;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50, v_RLC_Data1);

  f_Delay(0.06);     // NOTE 9
  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1, 

                                             cs_TimingInfo_Now, 

                                             15, 3, 1, 4); // NOTE 9
  // Step 25: Check: Does the UE transmit an RLC SDU  containing SDU#5 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU5 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 25");

  // Step 26: Check: Does the UE transmit an RLC SDU  containing SDU#6 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU6 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 26");

  // Step 27: Check: Does the UE transmit an RLC SDU  containing SDU#7 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU7 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 27");

  // Step 28: Check: Does the UE transmit an RLC SDU  containing SDU#8 in its data field ?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU8 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 28");

  // Step 28a: Send STATUS PDU upon Pollbit.

//  f_TxSTATUS_PDU(p_RLC_Rec);   // NOTE 6
  // Step 29: The SS transmits an RLC STATUS PDU to the UE.

  // This PDU acks PDUs up to those including SDU#7. ACK_SN=8.

  p_RLC_Rec.AM_VTA := 8;

  f_TxSTATUS_PDU ( p_RLC_Rec );

} // fl_EUTRA_7_2_3_15_TestBody


5 Execution Log Files

5.1 LG Electronics LEO3 UE

The LG electronics LEO3 UE passed this test case on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file tc_7_2_3_15_LG-log.html:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5.2 Samsung Tanum UE

The Samsung Tanum UE passed this test case on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file tc_7_2_3_15_Samsung-log.html:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

	[1]
	R5s100055: This archive comprises text format execution log file.

